
Matemática com Maple

O Maple é um sistema de álgebra computacional comercial de uso genérico. Constitui um ambiente
informático para a computação de expressões algébricas, simbólicas, permitindo o desenho de gráficos
a duas ou a três dimensões. Estes exerćıcios ajudam a começar a trabalhar com o programa.

O commando de linha básico

O sinal

>

indica que o Maple está à espera de input. Escreva os comandos directamente após este sinal e depois
prima a tecla enter. Por exemplo, a expressão 23 · 85 + 14/43 calcula-se com o comando

> 23*85 + 14/43;

o que gera o output

63310
43

Aritmética do Maple

No exemplo acima o Maple usou aritmética exacta; as respostas são números reais exactos. Porque
cada valor introduzido era um inteiro, o Maple assume que queremos ver aritmética exacta onde tal
seja posśıvel. Se quisermos respostas na forma decimal, uma maneira é usar a notação decimal na
expressão introduzida. Por exemplo, escrevendo 14.0 em vez de 14

> 23*85 + 14.0/43;

1955.325581

Para fazer cálculos sem mostrar o resultado, usa-se ”:” em vez de ”;”:

> A:=1-2/3: B:=1/3:

> A+B;

2
3

O śımbolo (%) refere o valor obtido no último cálculo.

> 1/3+%;

1

> 3/5-%;

−2
5

Para obter o resultado na forma decimal, usa-se a função evalf

> evalf(%);

−.4000000000

Pode especificar-se o número de d́ıgitos:

> evalf(1/3,20);

.33333333333333333333

> evalf(sqrt(2));

1.414213562

Exponenciação:

> 5^(1/2);

√
5

> 2^50;

1125899906842624

Algumas constantes:

> Pi; evalf(Pi);

π

3.141592654

> exp(1); evalf(exp(1));

e
2.718281828

> I^2;

−1

Mais alguma informação básica sobre o Maple

O valor de uma expressão pode ser dada por meio de uma fracção:

> (121/14-3^2)*11^2;

−605
14

Se pretender o resultado na forma decimal, basta usar evalf:

> evalf(%);

−43.21428571

Se pretende uma aproximação com um certo número de d́ıgitos:

> evalf(%,4);

−43.21

O número π em Maple:

> Pi;

π

Aproximações com 6 e 20 d́ıgitos, respectivamente:

> evalf(Pi,6); evalf(Pi,20);

3.14159
3.1415926535897932385

Outro exemplo:

> sqrt(3);

√
3

> evalf(sqrt(3),4);

1.732

Funções básicas:

> exp(x);

ex

> exp(2);

e2

> sin(Pi/2);

Somatórios(exemplos):

> sum(k*k, k=0..3);

14

> sum(2*k-1,k=1..50);

2500

> seq(sum(k,k=1..n), n=1..30);

1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91, 105, 120, 136, 153, 171, 190, 210, 231, 253, 276,

300, 325, 351, 378, 406, 435, 465

> sum(2*k-1,k=1..50);

2500

> sum(i,i=1..n);

(n + 1)2

2
− n

2
− 1

2

Polinómios

> (y+x*(y+z))^3;

(y + x(y + z))3

> expand(%);

y3 + 3xy3 + 3y2xz + 3x2y3 + 6x2y2z + 3yx2z2 + x3y3 + 3x3y2z + 3x3yz2 + x3z3

> collect(%,x);

(y3 + 3y2z + 3yz2 + z3)x3 + (3y3 + 6y2z + 3yz2)x2 + (3y3 + +3y2z)x + y3

> factor(%);

(y + xy + xz)3

Exerćıcio 1. Factorize e indique as ráızes do polinómio

x4 + x3 − 3x2 − x + 2

Um exemplo sobre juros. Pretende-se um modelo para calcular o capital acumulado a partir de uma
determinada quantia de dinheiro, sabendo que o juro é de 10% ao ano. Começando com um exemplo,
suponhamos que a quantia inicial é de 1000 euros.

> p(0)=1000;

p(0) = 1000

Para nos podermos referir a esta equação mais tarde vamos dar-lhe um nome, por exemplo, eq0.

> eq0:=%;

p(0) = 1000

Vamos atribuir também um nome à equação j = 0.1. (Note-se que p(0) + jp(0) é a quantia obtida no
final do primeiro ano de depósito.)

> eq1:=(j=.1);

eq1 := j = .1

Denotando por p(1) o total do depósito no final do primeiro ano, temos a equação:

> p(1)=p(0)+j*p(0);

p(1) = p(0) + jp(0)

Se quisermos saber o valor de p(1) para os valores que assumimos no ińıcio, basta usar as equações eq0
e eq1 para especificar esses valores, com o comando subs. A seguir ilustra-se a sequência dos cálculos:

A rotina subs permite obter valores para uma expressão envolvendo variáveis pondo

subs(variável1=certa coisa1, variável2=certa coisa2, ..., variáveln=certa coisan, expressão)

Usando esta rotina:

> subs(eq0,eq1,%);

p(1) = 1100.0

Exerćıcio 2. Use p(1) para determinar o capital em depósito após o primeiro ano, supondo que o capital
inicial foi de 2500 euros e o juro anual é de 6%.

Claro que agora nos interessa uma fórmula que permita calcular o capital ao fim de cada n anos. O
capital no final do ano n + 1 pode exprimir-se em função do capital no final do ano n por meio da
igualdade:

> p(n+1)=p(n)+i*p(n);

p(n + 1) = p(n) + ip(n)

Uma equação deste tipo é dita de recorrência. um cálculo com lápis e papel leva facilmente ao valor
de cada p(n) em função de p(0). Mas vamos usar o Maple para isso, dando simultaneamente um nome
à equação obtida.

> eqn:=p(n)=rsolve(%,p(n));

eqn := p(n) = p(0)(1 + i)n

Isto pode ser usado para criar uma função que, a partir dum dado capital inicial p(0), calcula o capital
que se tem ao fim de n anos. A fórmula para p(n) é:

> subs(eqn,p(n));

p(0)(1 + i)n

Uma função que calcula esta expressão em função de n e i é:

> f:=unapply(%,n,i);

f := (n, i) → p(0)(1 + i)n

Portanto, o capital após 10 anos à taxa de 18% é:

> f(10,.18);

5.233835554p(0)

Para p(0) = 1000 tem-se:

> subs(p(0)=1000,%);

5233.835554

Exerćıcio 3. Use a função f para determinar qual o capital que terá num certo banco daqui a 9 anos,
onde depositou 1500 euros no dia 31 de Dezembro de 2001 a uma taxa anual de 15%.

Iteração

(a) Iteração sobre uma progressão aritmética (for loop):

> for i from 5 to 11 by 3 do

> i*i;

> od;

25
64
121

(b) Iteração sobre os elementos de uma lista (for loop):

> mylist:=[5,8,11]:

> for i in mylist do

> i*i;

> od;

25
64
121

Alternativamente:

> mylist:={5,8,11}:

> for i in mylist do

> i*i;

> od;

25
64
121

Exerćıcio 4. Calcule o cubo dos números 2, 4, 8 e 248.

(c) Iterações mais gerais (while loop):

> i:=3:

> while i<=9 do

> print(i*i);

> i:=i+2;

> od:

9
25
49
81

> i:=1:

> while i^7 < 150 do

> print(i);

> i:=i+i^2;

> od:

1
2

Também se pode combinar o for com o while:

> for i from 3 to 44 while i^2<50 do

> i, i^2;

> od;

3, 9
4, 16
5, 25
6, 36
7, 49

Exerćıcio 5. Use um while loop para determinar todos os valores inteiros positivos n tais que n3 < 500.

Exerćıcio 6. Determine o valor de
√

n para todo o n de 1 a 10,

(a) usando um for loop.

(b) usando um while loop.

Exerćıcio 7. Use o Maple para resolver o Exerćıcio 8(d) da Folha TP-3 (página 2): n3 − n é diviśıvel
por 3 para qualquer n ∈ N, ou seja, o resto da divisão inteira de n3 − n por 3 é igual a zero para
qualquer n ∈ N.
A instrução do Maple para o cálculo do resto da divisão inteira de m por n é irem(m,n).

(a) Base indutiva: calcule

> irem(1^3-1,3);

(b) Hipótese indutiva: irem(n3 − n, 3)=0.

Passo indutivo: vamos calcular a diferença entre (n + 1)3 − (n + 1) e n3 − n:

> (n+1)^3-(n+1)-(n^3-n);

> expand(%);

A expressão que obteve representa claramente um múltiplo de 3. Como, por hipótese indutiva,
n3 − n é múltiplo de 3, conclui-se que (n + 1)3 − (n + 1) também é múltiplo de 3.

Funções e procedimentos

Uma função pode ser definida facilmente em Maple usando o operador −>. Para uma função de p

variáveis v1, v2, . . . , vp a sintaxe é:

(v1,v2, ...,vp)−> exp

onde exp é uma expressão escrita explicitamente como uma função das variáveis v1, v2, . . . , vp que
corresponde ao resultado dado pela função.
Por exemplo:

> f_1:=x->(x-1)^2:

> f_2:=(x,y)->xy:

> f_3:=(a,b,c)-> a and b or not c:

Podemos agora calcular o valor das funções para determinados valores das variáveis:

> f_1(7);

36

> f_2(2,-3);

−6

> f_3(true,true,false);

true

Podemos também aplicar a função a valores indeterminados:

> f_1(t);

(t− 1)2

> f_1(a/b);

(
a

b
− 1)2

Se for necessário ver qual a actual definição de f , usa-se o comando eval:

> eval(f_1);

x → (x− 1)2

Um procedimento consiste em várias instruções delimitadas pelas duas palavras reservadas proc e
end.

A seguir vamos usar procedimentos com a seguinte sintaxe simples:

proc(v1,v2,...vp)
<instrução1>;
<instrução2>;
...
<instruçãon>;
end

Um exemplo:

> f:=proc(x,y)

> if x<y then x+y

> else x*y fi;

> end:

Note que usamos um if, com a seguinte sintaxe:

if <condição1> then <expressão1>

else <expressão2> fi

Aqui fi está em vez de end if.

Usando o procedimento acima podemos agora calcular valores de f(x, y) :

> f(1,2);

3

> f(2,1);

2

Gráficos

Basta usar o comando plot:

> plot(x^2+3, x=0..4, y=0..10);

> plot({n,n*ln(n)}, n=2..6);

> plot({sin(x),cos(x)},x=-Pi..Pi);

> plot({floor(x), ceil(x)}, x=-10..10);

floor (função floor): caracteŕıstica de um número, i.e. o maior número inteiro que é menor ou igual
que o número dado; ceil (função ceiling): menor número inteiro que é maior ou igual do que o número.

Exerćıcio 8. Use os seus conhecimentos de Maple para averiguar rapidamente se:

(a) f : (0, 14)→(0, 14) definida por f(x) = (x2 + x + 1) mod 15 é injectiva. (Será sobrejectiva?)

(b) a relação constitúıda pelos pares (x, y) tais que x ≡7 y no conjunto {1, 3, 5, 7} é uma função.

(c) a função g : (0, 6)→(0, 6) definida por g(x) = x2 mod 7 é ou não injectiva. (Será sobrejectiva?)

(d) a função h : (0, 1000)→(0, 11) definida por h(x) = (x3 + x2 + x + 1) mod 11 é sobrejectiva.

Trace o gráfico da função caracteŕıstica no intervalo [−5, 5].

